
You’ve Sent Mail: Sending Reports Automatically Through SAS 

ABSTRACT 
We often have reports to produce that are ultimately sent out via electronic mail.  Rather than saving your 
output and manually emailing, learn how to automatically send your reports through SAS.  SAS has the 
capability to send emails with or without attachments to one or many people or even site specific reports via 
a macro. 
 
INTRODUCTION 
This paper discusses 3 separate uses for automatic emails.  First, an automatic email with no attachment 
can be sent notifying someone that a report is available on a website or shared drive.  Second, an automatic 
email with an attachment can be sent to multiple recipients.  Third, personalized site specific reports can be 
created and sent to specific recipients.  Examples for all 3 of these uses are shown. 
 
EXAMPLE 1:  SINGLE EMAIL WITH NO ATTACHMENTS 
The simplest use of SAS’ automatic emailing capability is to send one email to a recipient or recipients with 
no attachment.  This would be useful if you need to notify colleagues or clients that a report is available for 
viewing on a website or server. 
 
In example 1 below, the filename statement creates a temporary file with any name that you choose and 
must be 8 characters or less.  In this case it is called “myemail.”  The “EMAIL” specifies the device type, 
telling SAS that you want to send an email.  Next are the options, such as the address or addresses to 
which the email should be sent, the subject, an attachment, any CCs or BCCs.  The data _null_ step 
executes the “myemail” file and includes the text for the body of the message in the put statement. 
 
 
FILENAME myemail EMAIL from=("theresam@uab.edu") 
 to=("theresam@uab.edu" "tmcvie@ms.soph.uab.edu") 
 cc=("theresam@uab.edu") 
 Subject = "An automatic email sent from SAS"; 
 
data _null_; 
file myemail; 
put "Your report is now available online." 
/ / "Thank you and have a great day." 
/ / " " 
/ /"Sincerely," 
/ /"Theresa McVie" 
/ / " " 
/ / "This is an automated email sent by SAS on behalf of Theresa McVie"; 
run; 
 
EXAMPLE 2:  SINGLE EMAIL WITH AN ATTACHMENT 
Extending example 1, let’s attach the report to the email message rather than simply notifying the recipient it 
is ready.  This will revolutionize your life by allowing you to run a program via scheduled tasks as often as 
you would like without having to run the report and email it manually.  There is only a simple line of code 
(attach=) added to the filename statement. 
 
*Create data for report; 
data one; 
INPUT NAME $ 1-10 SEX $ 12 AGE Height WEIGHT; 
datalines; 

STEVE      M 41 74 170 
ROCKY      M 42 68 166 
KURT       M 39 72 167 
DEBORAH    F 30 66 124 
JACQUELINE F 33 66 115 
; 
run; 
 
*Create rtf file to be emailed via SAS; 
ods rtf file = "C:\Documents and Settings\report1_&SYSDATE..rtf"; 
proc freq data=one; 
title "First Example Report"; 
tables sex; 
run; 
ods rtf close; 
 
*Begin email code; 
FILENAME myemail2 EMAIL from=("theresam@uab.edu") 
 to=("tmcvie@ms.soph.uab.edu") 
 Subject = "An email with attachment sent from SAS" 
 Attach = "C:\Documents and Settings\report1_&SYSDATE..rtf"; 
 
data _null_; 
file myemail2; 
put "Please see attached for your daily report." 
/ / "Thank you and have a great day." 
/ / " " 
/ /"Sincerely," 
/ /"Theresa McVie" 
/ / " " 
/ / "This is an automated email sent by SAS on behalf of Theresa McVie"; 
run; 
 
EXAMPLE 3:  CREATE A REPORT “PERSONALIZED” FOR A PARTICULAR SITE’S DATA 
Now let’s create multiple files and send them to different recipients.  To accomplish this the SAS Macro 
facility will be used.  First, two datasets are created, one containing the email addresses and one with 
dummy data for the reports.  Next, a macro is used to create separate reports for each parent.  Finally, an 
additional macro is used to send the separate emails.  Rather than having to invoke the macros for each 
parent, CALL EXECUTE statements are used.  In addition to being simpler, this approach also greatly 
improves the flexibility of the program.   
 
 
*CREATE DATA FILES TO BE USED FOR REPORTS; 
data contact; 
length EMAIL $25; 
INPUT NAME $ 1-10 EMAIL $ @12; 
datalines; 
STEVE      Steve@example.com 
ROCKY      Rocky@example.com 
KURT       Kurt@example.com 
DEBORAH    Deborah@example.com 
JACQUELINE Jacqueline@example.com 

; 
run; 
 
data two; 
INPUT NAME $ 1-10 CHILD_NAME $ 12-23 CHILD_SEX $24 CHILD_AGE; 
datalines; 
STEVE      CAROLINE    F 5 
STEVE      LUKE        M 3 
ROCKY      MARTHA      F 7 
ROCKY      STEPHANIE   F 3 
ROCKY      JENNIFER    F 1 
KURT       LINDSAY     F 9 
KURT       SARAH       F 9 
DEBORAH    MARK        M 5 
DEBORAH    CHRISTOPHER M 3 
DEBORAH    WILLIAM     M 1 
JACQUELINE MARY        F 9 
JACQUELINE JOSH        M 7 
JACQUELINE SEAN        M 5 
; 
run; 
 
*CREATE PERSONALIZED REPORTS; 
%macro report(parent = ); 
ods rtf file = "C:\Documents and Settings\Report for &parent &SYSDATE..rtf"; 
PROC TABULATE data=two format=6.0; 
title "Statistics by Parent"; 
title2 "For &parent"; 
where name="&parent"; 
class CHILD_SEX NAME; 
var CHILD_AGE; 
table NAME,ALL CHILD_SEX*N CHILD_AGE*(MEAN*f=6.1 STD*f=6.1)/misstext="0"; 
run; 
ods rtf close; 
%mend report; 
 
*EXECUTE THE REPORT MACRO USING THE CONTACT LIST; 
data _NULL_; 
SET contact; 
call execute('%Report(PARENT = '||NAME||')'); 
RUN; 
 
*CREATE PERSONALIZED EMAILS; 
%MACRO EMAIL(PARENT =, EMAIL=); 
filename mymail email from="theresam@uab.edu" 
                      to=("&email") 
                      subject="Report for Parents" 
     attach = ("C:\Documents and Settings\Report 
for &parent &SYSDATE..rtf");  
data _null_; 
   file mymail; 

      put  // "Dear &parent,"  
  /  " " 
  /  "Please see attached file for the recent parent report." 
  /  " " 
  / "Thank you, " 
  /  " " 
  /   "Theresa McVie"; 
     
RUN; 
%MEND EMAIL; 
 
*EXECUTE EMAIL MACRO USING CONTACT LIST; 
data _NULL_; 
SET contact; 
call execute('%EMAIL(PARENT = '||NAME||', EMAIL ='||EMAIL||')'); 
RUN; 
 
Conclusion 
Imagine having to write hundreds of emails to send out reports individualized for people or groups of people.  
Emailing reports automatically in your SAS code will eliminate his manual step and save you a lot of time 
with a little effort up front to create your code.  This paper introduced you to the basics of this concept, 
various options, and specific uses.  Once you start using this convenient feature, you will find that there are 
many more uses of automatic emailing that one could apply. 

